
Experimental methods

The N-acetyl-[U-13C,15N,70%-17O]-L-valyl-L-leucine (N-Ac-VL) sample used here was 

prepared by 17O labeling U-13C,15N-FMOC-L-valine and U-13C,15N-FMOC-L-leucine with H2
17O 

via a multiple turnover reaction.1 The procedure involves reacting an FMOC, BOC, Trt, or OtBu 

amino acid with excess carbodiimide and in this case 70% H2
17O. The reaction is a kinetically 

enhanced multiple turnover process that provides the 17O-labeled FMOC amino acids in high yield 

and an isotopic enrichment equal to that of the starting H2
17O. It appears to be a generally 

applicable approach for 17O carboxyl groups. Further details of the procedure are available in Refs. 

1,2.

NMR experiments were acquired at 18.8 T on a Bruker Avance III HD console using a 

triple-resonance probe designed and constructed at the NHMFL using a JEOL 0.75 mm MAS 

stator. The carrier frequencies for 1H, 17O, and 15N were 800.12, 108.48, and 81.08 MHz, 

respectively. All experiments were performed at a spinning frequency ωr/2π = 90 kHz. In all 

instances, the initial 17O polarization was enhanced by saturation/inversion of the 17O satellite-

transitions using a WURST pulse3,4 with a sweep width equal to the MAS frequency of 90 kHz, 1 

ms pulse duration, peak rf field ω1/2π = 12.4 kHz, and an offset of +450 kHz, leading to 17O signal 

enhancement for the N-Ac-VL sample by a factor of approximately three. Other experimental 

parameters are given in Figs. S1-S3. The cosine low-power MQMAS sequence (cos-lpMQMAS)5,6 

with triple-/single-quantum split-t1 evolution was used to obtain isotropic 17O resolution using 

cosine pulses with a duration of two rotor periods, and a ratio k = 19/12 for S = 5/2 nuclei.7,8 For 1H-

17O heteronuclear dipolar recoupling during the RESPDOR9 portion of the ohNH{O} experiment, 

the SR42
1 sequence10 was used on the 1H channel at an rf field amplitude ω1/2π twice the spinning 

frequency, 180 kHz. The RESPDOR build-up curve was fit using the following equation11

4 )  

where J±1/4 are Bessel functions of the first kind, dHO is the 1H-17O dipolar coupling, and tmix is the 

total duration of SR42
1 dipolar recoupling. The dipolar coupling in units of hertz is defined as

𝑑𝐻𝑂 =‒
𝛾𝐻𝛾𝑂ℏ

2𝜋𝑟 3
𝐻𝑂

⋅ 10 ‒ 7

where γH (26.75222×107 rad·T-1·s-1) and γO (-36.264×107 rad·T-1·s-1) are the gyromagnetic ratios 

of the 1H and 17O nuclei, ħ (1.05457×10-34 J·s) is Planck’s constant divided by π, and rHO is the 

distance in meters between the H and O sites. The dipolar coupling between 15N and 17O nuclei 

can be calculated using the above equation by replacing the 1H related parameters with 15N ones, 

e.g., γN = -2.7116×107 rad·T-1·s-1.

1H chemical shifts were referenced indirectly to neat tetramethylsilane (δiso = 0 ppm) by 

setting the highest frequency peak of histidine⋅HCl⋅H2O to 17.2 ppm.12 15N and 17O shifts were 

referenced to liquid NH3 and D2O at 0 ppm, respectively, by using the reference frequency for 1H 

at 0 ppm and the IUPAC recommended standard frequency ratios.13 All spectra were processed 

using Bruker Topspin 4.1.1 and imported into MATLAB R2020b Update 6 (9.9.0.1718557) for 

analysis and plotting. The isotropic 17O dimension of 3D spectra, the axis was scaled and 

referenced according to the 'unified' convention for MQMAS and STMAS.14 Essentially, the 17O 

spectral window was divided by a factor (3 - k) = 17/12, while keeping the Larmor frequency and 

shift reference the same as for conventional 1D 17O spectra.

Fig. S1. Pulse sequence schematic and experimental parameters used for the three-dimensional 
OhNH experiment.

-------------------------------- OhNH pulse sequence in Bruker format --------------------------------
; OhNH.ih - written by ivan
;  - 3D H/N/O correlation under fast MAS
;  - O(aniso) -(CP)-> H -(CP)-> N -(CP)-> H

;parameters:
;ns : 4*n
;p1 : 1H 90 pulse at PL1
;p2 : 15N 90 pulse at PL2
;p3 : 17O 90 pulse at PL3
;p8 : ST-saturation WURST pulse at SP8
;p14 : 17O->1H contact time at SP5(1H) + PL9(17O)
;p15 : 1H->15N contact time at SP0(1H) + PL5(15N)
;p16 : 15N->1H contact time at SP1(1H) + PL6(15N)
;p27 : duration of water suppression (MISSISSIPI)
;pcpd1 : =25us for 10kHz waltz16
;pcpd2 : =25us for 10kHz waltz16
;d0 : initial t2 delay
;d1 : recycle delay
;d20 : initial t1 delay
;in0 : = t2(15N) increment
;in20 : = t1(17O) increment

;pl1 : 1H pulse power level
;pl2 : 15N pulse power level
;pl3 : 17O pulse power level
;pl5 : 15N CP(1H->15N) power level
;pl6 : 15N CP(15N->1H) power level
;pl9 : CP(17O->1H) power level
;pl11 : 1H dec power level
;pl12 : 15N dec power level
;pl17 : 1H WS power level
;sp0 : 1H CP(1H->15N) power level
;sp1 : 1H CP(15N->1H) power level
;sp5 : 1H CP(1H->17O) power level
;sp8 : ST-saturation WURST power level
;spnam8 : WURST shape, linear frequency sweep
;spoffs8 : ST-saturation WURST offset
;cpdprg1 : waltz16 at PL11
;cpdprg2 : waltz16 at PL12
;l30 : # rotor periods for WURST pulse
;$TYPE=ivan

"p8=1s*l30/cnst31"
"d0=0"
"d20=0"

aqseq 312

1 ze
2 10m do:f2
  d1

  (p8:sp8 ph20):f3  ;WURST ST saturation pulse
  3u

  5u pl1:f1 pl5:f2 pl3:f3
  (p3 ph1):f3  ;17O 90 pulse

  d20  ; 17O t1 evolution

  (p14:sp5 ph22):f1 (p14 ph20 pl9):f3  ;17O->1H CP

  (p15:sp0 ph20):f1 (p15 ph2 pl5):f2  ;1H->15N CP
  1u pl11:f1
  (1u cpds1):f1
  (center (d0):f2 (p3*2 ph20 pl3):f3)  ;17O pi-pulse @ mid t2
  2u
  (ralign (1u do):f1 (p2 ph20 pl2):f2)  ;15N 90 for z-filter

  (p27*0.25 ph20 p217):f1  ;MISSISSIPI
  (p27*0.25 ph21):f1
  (p27*0.25 ph22):f1
  (p27*0.25 ph23):f1
  (p2 ph3):f2  ;15N 90

  (p16:sp1 ph4):f1 (p16 ph20 pl6):f2  ;15N->1H CP
  0.5u pl12:f2
  go=2 ph31 cpds2:f2 finally do:f2
  1u do:f2

  10m mc #0 to 2
    F1PH( calph(ph1,+90), caldel(d20,+in20) )
    F2PH( calph(ph2,+90), caldel(d0,+in0) )
exit

ph1= 1 3
ph2= 1
ph3= 1 1 3 3
ph4={1}*4 {3}*4 {0}*4 {2}*4
ph20=0
ph21=1
ph22=2
ph23=3
ph31={0 2 2 0}^2^3^1
-------------------------------- OhNH pulse sequence in Bruker format --------------------------------

Fig. S2. Pulse sequence schematic and experimental parameters used for the three-dimensional 
O̅hNH experiment.

-------------------------------- O̅hNH pulse sequence in Bruker format --------------------------------
; OihNH.ih – written by ivan
;  - 3D H/N/O correlation under fast MAS
;  - O(iso) –(CP)-> H –(CP)-> N –(CP)-> H

;parameters:
;ns : 24*n
;p1 : 1H 90 pulse at PL1
;p2 : 15N 90 pulse at PL2
;p3 : 17O 90 pulse at PL3
;p7 : CT<->3Q conversion pulse at SP7
;p8 : ST-saturation WURST pulse at SP8
;p14 : 17O->1H contact time at SP5(1H) + PL9(17O)
;p15 : 1H->15N contact time at SP0(1H) + PL5(15N)
;p16 : 15N->1H contact time at SP1(1H) + PL6(15N)
;p27 : duration of water suppression (MISSISSIPI)
;pcpd1 : =25us for 10kHz waltz16
;pcpd2 : =25us for 10kHz waltz16
;d0 : initial t2 delay
;d1 : recycle delay
;d7 : =k*t1 delay for isotropic 17O evolution
;d20 : initial t1 delay
;in0 : =t2(15N) increment
;in7 : =k*t1 increment for isotropic 17O evolution
;in20 : =rotor period, t1(17O) increment
;pl1 : 1H pulse power level
;pl2 : 15N pulse power level
;pl3 : 17O pulse power level
;pl5 : 15N CP(1H->15N) power level
;pl6 : 15N CP(15N->1H) power level
;pl9 : CP(17O->1H) power level
;pl11 : 1H dec power level
;pl12 : 15N dec power level
;pl20 : 17O saturation pulse power level
;pl27 : 1H WS power level
;sp0 : 1H CP(1H->15N) power level
;sp1 : 1H CP(15N->1H) power level
;sp5 : 1H CP(1H->17O) power level
;sp7 : CT<->3Q conversion power level
;sp8 : ST-saturation WURST power level
;spnam7 : COS shape, cosine amplitude-modulated pulse
;spnam8 : WURST shape, linear frequency sweep
;spoffs8 : ST-saturation WURST offset
;cpdprg1 : waltz16 at PL11
;cpdprg2 : waltz16 at PL12
;cnst31 : MAS frequency [Hz]
;l30 : # rotor periods for WURST pulse
;$TYPE=ivan

"p7=2s/cnst31"
"p8=1s*l30/cnst31"
"d0=0"
"d20=0"
"d7=(2s/cnst31)*1.583333-p3"
"in7=in20*1.583333" ;k for S=5/2

aqseq 312

1 ze
2 10m do:f2
  d1



8

  (p8:sp8 ph20):f3  ;WURST ST saturation pulse
  3u

  5u pl1:f1 pl5:f2 pl3:f3
  (p3 ph1):f3  ;17O 90 pulse
  (p7:sp7 ph1):f3  ;17O 1Q->3Q
  d20  ; 17O t1 evolution
  (p7:sp7 ph20):f3  ;17O 3Q->1Q
  (p3*2 ph20 pl3):f3  ;17O 180 pulse
  d7

  (p14:sp5 ph22):f1 (p14 ph20 pl9):f3  ;17O->1H CP

  (p15:sp0 ph20):f1 (p15 ph2 pl5):f2  ;1H->15N CP
  1u pl11:f1
  (1u cpds1):f1
  (center (d0):f2 (p3*2 ph20 pl3):f3)  ;17O pi-pulse @ mid t2
  2u
  (ralign (1u do):f1 (p2 ph20 pl2):f2)  ;15N 90 for z-filter

  (p27*0.25 ph20 pl27):f1  ;MISSISSIPI
  (p27*0.25 ph21):f1
  (p27*0.25 ph22):f1
  (p27*0.25 ph23):f1
  (p2 ph3):f2  ;15N 90

  (p16:sp1 ph4):f1 (p16 ph20 pl6):f2  ;15N->1H CP
  0.5u pl12:f2
  go=2 ph31 cpds2:f2 finally do:f2
  1u do:f2
  10m mc #0 to 2
    F1PH( calph(ph1,+30), caldel(d20,+in20)&caldel(d7,+in7) )
    F2PH( calph(ph2,+90), caldel(d0,+in0) )
exit

ph1=(12) 0 2 4 6 8 10
ph2= 1
ph3={1}*6 {3}*6
ph4={1}*12 {3}*12 {0}*12 {2}*12
ph20=0
ph21=1
ph22=2
ph23=3
ph31={{0 2 0 2 0 2}^2}^2^3^1
-------------------------------- O̅hNH pulse sequence in Bruker format --------------------------------

Fig. S3. Pulse sequence schematic and experimental parameters used for the three-dimensional 
ohNH{O} experiment.

-------------------------------- OhNH{O} pulse sequence in Bruker format --------------------------------
; OhNHo.ih - written by ivan
;  - 3D H/N/O correlation under fast MAS
;  - pseudo-3D hNH spectra with 17O/1H RESPDOR in the 3rd dimension
;  - SR4 1H/17O dipolar recoupling

;parameters:
;ns : 32*n
;p1 : 1H 90 pulse at PL1
;p2 : 15N 90 pulse at PL2
;p3 : 17O 90 pulse at PL3
;p8 : ST-saturation WURST pulse at SP8
;p14 : 17O->1H contact time at SP5(1H) + PL9(17O)
;p15 : 1H->15N contact time at SP0(1H) + PL5(15N)
;p16 : 15N->1H contact time at SP1(1H) + PL6(15N)
;p20 : 17O RESPDOR saturation pulse at PL20
;p21 : 1H SR4 pulse at PL10, =tr/4
;p27 : duration of water suppression (MISSISSIPI)
;pcpd1 : =25us for 10kHz waltz16
;pcpd2 : =25us for 10kHz waltz16
;d0 : initial t2 delay

;d1 : recycle delay
;in0 : =t2(15N) increment
;pl1 : 1H pulse power level
;pl2 : 15N pulse power level
;pl3 : 17O pulse power level
;pl5 : 15N CP(1H->15N) power level
;pl6 : 15N CP(15N->1H) power level
;pl9 : CP(17O->1H) power level
;pl10 : =2vr, 1H SR4 power level
;pl11 : 1H dec power level
;pl12 : 15N dec power level
;pl20 : 17O saturation pulse power level
;pl27 : 1H WS power level
;sp0 : 1H CP(1H->15N) power level
;sp1 : 1H CP(15N->1H) power level
;sp5 : 1H CP(1H->17O) power level
;sp8 : ST-saturation WURST power level
;spnam8 : WURST shape, linear frequency sweep
;spoffs8 : ST-saturation WURST offset
;cpdprg1 : waltz16 at PL11
;cpdprg2 : waltz16 at PL12
;cnst20 : # rotor periods for 17O RESPDOR saturation pulse
;cnst31 : MAS frequency [Hz]
;l8 : L8*2 = # SR4 rotor periods to increment mixing time per t1
;l10 : initial # SR4 rotor periods
;l30 : # rotor periods for WURST pulse
;$TYPE=ivan

"p8=1s*l30/cnst31"
"p20=1s*cnst20/cnst31"
"p21=0.25s/cnst31"

"d0=0"
"d25=(1s/cnst31)"
"d26=(0.25s/cnst31)"
"d28=(1s/cnst31)-0.5u"

"l0=0"
"l2=0"
"l8=l10"
"cnst6=0"
"cnst8=0"

aqseq 312

1 ze
2 10m do:f2

  5u rpp16
  5u rpp17
  5u rpp18
  5u rpp19

"cnst6=180*trunc(((l2)/16)%2)"
  5u ip16+cnst6
  5u ip17+cnst6
  5u ip18+cnst6

  5u ip19+cnst6

  d1

  (p8:sp8 ph20):f3  ;WURST ST saturation pulse
  3u

  5u pl1:f1 pl5:f2 pl3:f3
  (p3 ph1):f3  ;17O 90 pulse

  (p14:sp5 ph22):f1 (p14 ph20 pl9):f3  ;17O->1H CP

  (p15:sp0 ph20):f1 (p15 ph2 pl5):f2  ;1H->15N CP
  1u pl11:f1
  (1u cpds1):f1
  (center (d0):f2 (p3*2 ph20 pl3):f3)  ;17O pi-pulse @ mid t2
  2u
  (ralign (1u do):f1 (p2 ph20 pl2):f2)  ;15N 90 for z-filter

  (p27*0.25 ph20 pl27):f1  ;MISSISSIPI
  (p27*0.25 ph21):f1
  (p27*0.25 ph22):f1
  (p27*0.25 ph23):f1
  (p2 ph3):f2  ;15N 90

  (p16:sp1 ph4):f1 (p16 ph20 pl6):f2  ;15N->1H CP

if "l2%2==0"{
     d25 pl20:f3
}
if "l2%2==1"{
     d25 pl=0[Watt]:f3
}
15 (p21 ph16 pl10):f1
   (p21 ph17):f1
   (p21 ph16^):f1
   (p21 ph17^):f1
   lo to 15 times l10
  d26
  (center (p1*2 ph5 pl1):f1 (p20 ph20):f3)
  d26
25 (p21 ph18 pl10):f1
   (p21 ph19):f1
   (p21 ph18^):f1
   (p21 ph19^):f1
   lo to 25 times l10
  d28

  0.5u pl12:f2
  go=2 ph31 cpds2:f2 finally do:f2
  1u do:f2
  10m mc #0 to 2
    F1PH( calph(ph2,+90), caldel(d0,+in0) )
    F2PH( calclc(l2,+1), calclc(l10,+l8) )
exit

ph1= 1 3

ph2= 1
ph3= 1 1 3 3
ph4= 1
ph5={0}*4 {1}*4 {2}*4 {3}*4
ph16=(360) 90 270 210 30 330 150
ph17=(360) 270 90 30 210 150 330
ph18=(360) 90 270 210 30 330 150
ph19=(360) 270 90 30 210 150 330
ph20=0
ph21=1
ph22=2
ph23=3
ph31={0 2 2 0}^2
-------------------------------- OhNH{O} pulse sequence in Bruker format --------------------------------

